

```

name          Buss;
Partno       TMD8000;
Revision     01;
Date        8/20/2006;
Designer    Len Bayles;
Company     TMD Innovations;
Location    None;
Assembly   None;
Device     p22v10;

```

```

/*****/
/*
/*      Buss control logic for 8008 computer
/*
/*
/*****/

```

```

/*
* Inputs:
*/

```

```

Pin 1 = S0;
Pin 2 = S1;
Pin 3 = S2;
Pin 4 = SYNC;
Pin 5 = D6;
Pin 6 = D7;
Pin 7 = O2;
Pin 8 = BBST;
Pin 9 = EXLL;
Pin 10 = EXLH;
Pin 11 = EXWR;
Pin 13 = PINT;

```

```

/*
* Outputs:
*/

```

```

Pin 23 = DBD;
Pin 22 = ALL;
Pin 21 = ALH;
Pin 20 = MER;
Pin 19 = MEW;
Pin 18 = IOR;
Pin 17 = IOW;
Pin 16 = ICLR;
Pin 15 = TLI;
Pin 14 = STOP;

```

```

/*
* Internal Logic:
*/

```

```

t1 = !S0 & S1 & !S2;
t1i = !S0 & S1 & S2;
t2 = !S0 & !S1 & S2;
t3 = S0 & !S1 & !S2;
stop = S0 & S1 & !S2;

```

```

pci = !D6 & !D7;          /* Memory Read Instruction */
pcr = !D6 & D7;          /* Memory Read Additional */
pcc = D6 & !D7;         /* I/O Operation */
pcw = D6 & D7;          /* Memory Write */

```

```

/*
* Logic:
*/

```

```

DBD = (t1 # t1i # t2 # t3) & !SYNC;          /* Data buss control - OUT */

ALL = ((t1 # t1i) & !SYNC & O2) # (EXLL & stop); /* Adress Latch LOW */
ALH = (t2 & !SYNC & O2) # (EXLH & stop);      /* Adress Latch HIGH */

MER = !((t3 & (pci # pcr) & SYNC) # (stop & !BBST)); /* Memory Read */
MEW = !((t3 & pcw & !SYNC & O2) # (EXWR & stop)); /* Memory Write */

IOR = t3 & pcc & SYNC;          /* IO Read */
IOW = t3 & pcc & !SYNC & O2;    /* IO Write */

ICLR = !(t1i & !SYNC);         /* INT CLR */

TLI = !t1i;                   /* TLI Sate - Intruppted */
STOP = stop;                  /* Stopped State */

```